

Web Ready Reporting of Student Projects

Suzanne L. Cross, The Pennsylvania State University, University Park PA

ABSTRACT

Student ratings of team projects are becoming an integral part of course assessment. SAS® can be used to import data from a variety of sources, create a nice report and save the instructor a lot of time. The students' ratings can be recorded on machine-readable forms or a handheld device and exported to SAS on a personal computer. The TABULATE procedure and the Output Delivery System are used to create a report ready to display on the course website.

INTRODUCTION

The impetus for this application came from the faculty teaching a senior-level projects course in which teams of students are evaluated by their peers. Each student used a rating form to score the other teams on a scale of 1 to 5 on ten factors. The instructor compiled the data by entering the scores from forms into spread sheets, calculating mean scores, two subscores and a total score for each team. This process was tedious and led the instructor to seek a more efficient method of evaluating the teams and reporting the results.

MAKE IT EASY!

Although collecting the ratings using a hand-held device is an interesting idea, I opted for using a machine-readable form to record the factor ratings. After class, the instructor delivers the forms for processing, and the ratings recorded by each student are input into SAS in a data step.

The Output Delivery System directs two reports to HTML files. The MEANS procedure is used to calculate and output the factor means, and create report1. The factor means are rescaled on a scale of 1-10 and two subscores are calculated as the sum of means in a data step. The TABULATE procedure is used to present report2 with the subscores highlighted. The tables are e-mailed to the instructor for posting to the course website within minutes.

THE RATING FORM

The rating form pictured is designed to overprint the factor descriptions on the top quarter of the form so the factors can be easily changed. In class, each student reviewer records the ten factor scores for each of the other team presentations by filling in the circles. Once processed, the resulting file contains one record per reviewer and can be read into SAS in the following data step.

```
filename scanned 'E:\NESUG2002\data02';
```

```
Data responses;
infile scanned missover;
input SequenceNumber 6-9 @20 @;
do Team=1 to 8;
input (Factor1-Factor10) (1.0)@;
output;
end;
run;
```

THE SUMMARY STATISTICS

The factor means can be calculated in the TABULATE procedure, but because the two subscores labeled Content and Writing are sums of means instead of the mean of the sum, I use the MEANS procedure to calculate the means first, and output them to the data set stats. The table of summary statistics by team is directed to the HTML table in the file named report1 (not shown here). I like to write reports to HTML files because they can be easily e-mailed, posted on a website, or opened with a browser without knowing what system the recipient uses.

```
ods listing close;
```

```
/* Create HTML table of stats for each factor */
ods html file="e:\NESUG2002\report1.html"
style=D3D;
```

```
/* Output factor means for each team */
proc means data=responses maxdec=2 nonobs;
class Team;
var Factor1-Factor10;
output out=stats (where=(type=1))
mean(Factor1-Factor10)=Factor1-Factor10;
title1 "Factor Statistics by Team";
run;
```

THE CALCULATIONS

The means for each factor are doubled here because the instructor wants to report the ratings adding-up to a total possible score of 100. Content Factor and Writing Factor are calculated as the sum of means.

```
/* Calculate subscores as the sum of factor */
/*      means on a scale of 1-10      */
data calculate;
set stats (keep=team Factor1-Factor10);
array Factors(10) Factor1-Factor10;
do i=1 to 10;
  Factors(i)=Factors(i)*2; /*double each rating*/
end;
ContentFactor=
  Factor1+Factor2+Factor3+Factor4+Factor5;
WritingFactor=
  Factor6+Factor7+Factor8+Factor9+Factor10;
Total=ContentFactor+WritingFactor;
run;
```

THE REPORT

The TABULATE procedure is used to present the report because it is easy to make it look good by changing default fonts, colors, & titles. The HTML file named report2 is attached in an email to the instructor who may post it to his course website for students to view.

```
/* Write table to html file */
ods html file="e:\NESUG2002\report2.html"
  style=D3D;

proc tabulate data=calculate;
  class team;
  var Factor1-Factor10 Total;
  var ContentFactor WritingFactor
    /style={background=yellow
      foreground=black};
  table
  (
    Factor1 Factor2 Factor3 Factor4 Factor5
    ContentFactor
    *{style={background=yellow
      font_weight=bold}}
    Factor6 Factor7 Factor8 Factor9 Factor10
    WritingFactor
    *{style={background=yellow
      font_weight=bold}} Total
  ),(Team*Sum='')
  /box="Rating Factor";
run;

ods html close;
ods listing;
```

Rating Factor	TEAM						
	1	2	3	4	5	6	7
Factor1	7.74	7.42	8.42	8.76	8.09	8.70	7.83
Factor2	6.52	7.25	7.83	7.81	7.83	8.81	7.22
Factor3	6.97	7.92	7.67	8.29	8.78	9.04	8.43
Factor4	8.43	8.25	8.00	7.62	8.43	9.74	7.74
Factor5	7.83	8.33	8.67	9.33	8.00	8.70	7.91
ContentFactor	37.39	39.17	40.68	41.81	41.13	44.78	39.13
Factor6	8.81	8.92	8.42	9.05	8.87	8.78	8.35
Factor7	8.35	8.00	8.25	8.76	8.43	8.43	8.52
Factor8	8.26	8.00	8.92	8.95	9.00	9.22	7.13
Factor9	8.09	7.67	8.42	8.76	8.17	8.17	8.19
Factor10	7.65	8.33	8.08	8.76	8.17	8.70	7.13
WritingFactor	40.96	40.92	42.08	44.29	42.65	43.30	39.32
Total	78.35	80.08	82.67	86.10	83.78	88.08	78.45

CONCLUSION

While this application is not impressive, it is significant because it is an example of using SAS to enhance other technologies. Using SAS to improve data analysis and reporting at the Pennsylvania State University adds value to the reports that professors and students receive, and is flexible enough to handle just about any data source or analysis that comes up.

In the future, I'd like to use palm-held devices to record the ratings and transfer the data to a personal computer running SAS for reporting.

ACKNOWLEDGEMENTS

SAS® is a registered trademark of SAS Institute, Inc.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Sue Cross
The Pennsylvania State University
23 Willard Building
University Park PA 16802
(814) 863-2598
slc@psu.edu
[http:// www.personal.psu.edu/~slc](http://www.personal.psu.edu/~slc)